# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=python-transaction
pkgver=3.1.0
pkgrel=2
pkgdesc='Transaction management for Python'
arch=('any')
license=('ZPL')
url='https://github.com/zopefoundation/transaction'
depends=('python-zope-interface')
makedepends=('python-setuptools')
checkdepends=('python-zope-testrunner' 'python-mock')
source=("https://github.com/zopefoundation/transaction/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha512sums=('44a1c6e4a9790953a4ca31eb9660d85723dca37694a38c80ccc09945b1e23886c4ca2f0ca6dd705d489f80c068ecf921e42a9e0f1a6958b50b7d0e7074040548')

build() {
  cd transaction-$pkgver
  python setup.py build
}

check() {
  cd transaction-$pkgver
  PYTHONPATH="$PWD"/build/lib zope-testrunner --test-path=src
}

package() {
  cd transaction-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1
}

# vim:set ts=2 sw=2 et:
